

Borel Resummation of Divergent Series

Using Gfun

Frédéric Chyzak, Marianne Durand, and Bruno Salvy

June, 2001

Abstract: We expand on ideas of Balser, Lutz, and Schäfke by showing how coefficients and integrals involved in calculations related to the analytic continuation of Borel transforms obey simple recurrences that lead to efficient numerical computations. This work is a follow-up to a talk by Donald Lutz at our Algorithms seminar, and summarized in [Durand].

1. Borel-Laplace resummation and Euler acceleration

Starting with a linear differential equation with polynomial coefficients satisfied by a formal power series

$$x^{\sim} = \sum_{n=0}^{\infty} x_n z^n,$$

it is possible to compute a differential equation satisfied by the Borel transform of $x^{\sim}$. We assume that $x^{\sim}$ is Gevrey 1, which means that $x_n \leq A c^n n!$ for some constants A and c , so that the Borel transform of $x^{\sim}$ defined by

$$y(z) = \sum_{n=0}^{\infty} \frac{x_n z^n}{n!}$$

is convergent on some neighbourhood of the origin. The Borel transform has an "inverse", the Laplace transform defined by

$$\text{Laplace}(y) = \int_0^{\infty} \mathbf{e}^{\left(-\frac{t}{z}\right)} y(t) dt.$$

Provided this integral converges, the function it defines has $z x^{\sim}$ for expansion as $z \rightarrow 0 +$. Then applying the change of variable $t = \phi(\alpha)$ to the integral computes the acceleration "a la Euler" [Lutz et al.]

$$z x^{\sim} = \int_0^{\infty} \mathbf{e}^{\left(-\frac{t}{z}\right)} y(\phi(\psi(t))) dt.$$

We note

$$y(\phi(z)) = \sum_{n=0}^{\infty} d_n z^n \text{ and } q_n(z) = \int_0^{\infty} \mathbf{e}^{\left(-\frac{t}{z}\right)} \psi(t)^n dt,$$

where ψ is the functional inverse of the rational function ϕ . In terms of formal power series, the product $z x^{\sim}$ equals the Taylor expansion of

$$\sum_{n=0}^{\infty} q_n d_n,$$

where the integrals q_n are independent of x .

This process is illustrated in the present worksheet using a simple mapping function ϕ on the double confluent Heun equation. The Heun equation is the generic differential equation with four regular singular points located at 0, 1, c , and ∞ , see [DuLoRi92]. The double confluent Heun equation is obtained by letting the singularity located at c tend to the one located at ∞ , and the singularity located at 1 tend to 0. The equation obtained then has two irregular

singular points located at 0 and ∞ . The example we study is the double confluent Heun equation in the form

$$\begin{aligned} > \text{heun_infp} := z^2 \left(\frac{\partial^2}{\partial z^2} f(z) \right) + (z + \alpha z^2 + \alpha) \left(\frac{\partial}{\partial z} f(z) \right) \\ &+ \frac{(2 \alpha z^2 \beta_1 + \alpha z^2 + \alpha^2 z - 2 \gamma z + 2 \alpha \beta_1 - \alpha) f(z)}{2 z} \end{aligned}$$

`> readlib(gfun) :`

Since the point of interest is infinity and the [gfun](#) package works at the origin, we first change the variable (using [gfun](#)):

`> heunp:=gfun[algebraicsubs](heun_infp, z*f-1, f(z));`

$$\text{heunp} := (-2 z^2 \alpha \beta_1 + \alpha z^2 - \alpha^2 z + 2 \gamma z - 2 \alpha \beta_1 - \alpha) f(z)$$

$$+ (-2 z^2 + 2 z^3 \alpha + 2 z \alpha) \left(\frac{\partial}{\partial z} f(z) \right) - 2 z^3 \left(\frac{\partial^2}{\partial z^2} f(z) \right)$$

We concentrate on the following set of special values for the parameters:

$$> \text{paramform} := \left[\alpha = -1, \beta_{-1} = \frac{1}{2}, \beta_1 = \frac{1}{2}, \gamma = \frac{1}{3} \right]$$

Substitution of these parameters in the differential equation gives

> `heun := subs (paramform, heunp) ;`

$$\text{heun} := \left(-\frac{1}{3} z + 2 \right) f(z) + (-2 z^2 - 2 z^3 - 2 z) \left(\frac{\partial}{\partial z} f(z) \right) - 2 z^3 \left(\frac{\partial^2}{\partial z^2} f(z) \right)$$

From this equation, we obtain a recurrence equation for the Taylor series coefficients

> `recheunseries := gfun[diffeqtorec] (heun, f(z), u(n)) ;`

`recheunseries := {`

$$-6 n u(n) + (-7 - 12 n - 6 n^2) u(n + 1) + (-6 - 6 n) u(n + 2),$$
`u(0) = 0 }`

This recurrence yields an efficient procedure to evaluate the coefficients recursively:

> `heundiv := gfun[rectoproc] (recheunseries union {u(1)=1/2}, u(n), remember) ;`

`heundiv := proc(n)`

option remember;

$$\begin{aligned} & - (12*\text{procname}(n-2) - 7*\text{procname}(n-1) + (\\ & -6*\text{procname}(n-2) + 12*\text{procname}(n-1) \\ & - 6*\text{procname}(n-1)*n)*n) / (6 - 6*n) \end{aligned}$$

end

(1) = 1 / 2

(0) = 0


From this procedure, the divergence is clear from the growth of the first coefficients:

> seq(heundiv(i), i=1..15);

$$\begin{aligned} & \frac{1}{2}, \frac{-7}{12}, \frac{139}{144}, \frac{-6637}{2592}, \frac{598753}{62208}, \frac{-86588791}{1866240}, \frac{18250889947}{67184640}, \\ & \frac{-5271793461229}{2821754880}, \frac{285349416100951}{19349176320}, \frac{-960609947353470343}{7313988648960}, \\ & \frac{571502026077983064331}{438839318937600}, \frac{-412023777148221194533837}{28963395049881600}, \\ & \frac{353911104407615639043543169}{2085364443591475200}, \\ & \frac{-357083839712993494956732905527}{162658426600135065600}, \\ & \frac{418134484182977410016133713165083}{13663307834411345510400} \end{aligned}$$

The generating function associated with the sequence $\text{heundiv}(n)$ is divergent. It is possible here to approach the corresponding function using summation to the least term. This method consists in summing the terms $\text{heundiv}(n) z^n$ up to the smallest one. Under certain conditions that are fulfilled here, the total differs from the function by the value of this smallest term.

```
> calculheundiv := proc(heundiv, z)
    local total, previous, last, n;
    previous := heundiv(1)*z;
    total := previous;
    last := heundiv(2)*z^2;
    for n from 3 while abs(last) < abs(previous) do
        total := total + last;
        previous := last;
        last := heundiv(n)*z^n
    od;
    userinfo(1, 'heundiv', n, last);
    evalf(total)
end
> plot('calculheundiv'(heundiv, z), z=0..0.3);
```



As z tends to infinity, the imprecision of this summation grows quickly.

2. Differential equation satisfied by the Borel transform

The class of solutions of linear differential equations enjoys many closure properties which are implemented in the [gfun](#) package for the case of equations with polynomial coefficients. One of them is closure under Borel transform. Here is the differential equation satisfied by the Borel transform of the divergent solution of the Heun equation:

```
> bdeqp:=op(select(has,gfun[borel](heunp,f(z), 'diffeq'),z));
```

$$bdeqp := (2 \alpha \beta_{-1} - \alpha) f(z) + (-2 z \alpha + \alpha^2 + 2 - 2 \gamma) \left(\frac{\partial}{\partial z} f(z) \right) \\ + (6 z + 2 \alpha \beta_1 - 3 \alpha) \left(\frac{\partial^2}{\partial z^2} f(z) \right) + (2 z^2 - 2 z \alpha) \left(\frac{\partial^3}{\partial z^3} f(z) \right)$$

and the equation specialized at the parameters

```
> bdeq:=subs(paramform,bdeqp):
```

From this follows the recurrence satisfied by the Taylor coefficients of the Borel transform:

```
> collect(gfun[diffeqtorec](bdeq,f(z),a(n)),a,factor);
```

$$-\alpha(-2 \beta_{-1} + 1 + 2 n) a(n) \\ + (n + 1)(2 n^2 + 4 n + \alpha^2 + 2 - 2 \gamma) a(n + 1) \\ - \alpha(n + 1)(n + 2)(2 n - 2 \beta_1 + 3) a(n + 2)$$

Check:

```
> l15:=gfun[rectoproc]({subs(paramform,%),a(0)=0,a(1)=1/2},a(n),list)(15);
```

$$\begin{aligned}
l15 := & \left[0, \frac{1}{2}, \frac{-7}{24}, \frac{139}{864}, \frac{-6637}{62208}, \frac{598753}{7464960}, \frac{-86588791}{1343692800}, \right. \\
& \frac{18250889947}{338610585600}, \frac{-5271793461229}{113773156761600}, \frac{285349416100951}{7021429103001600}, \\
& \frac{-960609947353470343}{26541002009346048000}, \frac{571502026077983064331}{17517061326168391680000}, \\
& \frac{-412023777148221194533837}{13873512570325366210560000}, \\
& \frac{353911104407615639043543169}{12985607765824542773084160000}, \\
& \frac{-357083839712993494956732905527}{14180283680280400708207902720000}, \\
& \left. \frac{418134484182977410016133713165083}{17867157437153304892341957427200000} \right]
\end{aligned}$$

This list corresponds to the list above, the k th element being scaled by $\frac{1}{k!}$.

This differential equation and this recurrence can be used to compute (but not necessarily fast) the analytic continuation of the Borel transform.

3. The coefficients of the composition with an algebraic function satisfy a linear

recurrence

This is another closure property of solutions of linear differential equation that we exploit here. The coefficients d_n are defined by

$$y(\phi(z)) = \sum_{n=0}^{\infty} d_n z^n$$

where ϕ is a rational function and y is defined by

$$y(z) = \sum_{n=0}^{\infty} \frac{x_n z^n}{n!}.$$

On the example of the Heun equation, with $\phi = \frac{1}{(1-t)^2} - 1$ we obtain

```
> eq := (1 - z)^2 (f + 1) - 1
> dneqp := gfun[algebraicsubs] (bdeqp, eq, f(z));
dneqp := (4 α - 8 α β1) f(z) + (-4 + 4 γ + 6 α β1 + 48 z
- 6 α2 z2 - 267 α z2 - 9 α + 6 α2 z - 12 γ z - 198 z2
+ 60 α z2 β1 + 12 γ z2 + 469 z3 α + 83 z α + 406 z3 - 30 α β1 z
- 495 z4 α + 309 z5 α - 105 z6 α + 2 α2 z3 + 15 z7 α - 465 z4
+ 303 z5 - 105 z6 + 15 z7 - 60 α z3 β1 - 4 γ z3 - 6 α z5 β1
+ 30 z4 α β1 - 2 α2) (∂/∂z f(z)) + (-2 α β1 - 12 z + 162 α z2
+ 3 α + 90 z2 - 30 α z2 β1 - 384 z3 α - 36 z α - 276 z3
```

$$\begin{aligned}
& + 12 \alpha \beta_1 z + 540 z^4 \alpha - 468 z^5 \alpha + 246 z^6 \alpha + 9 z^8 \alpha + 9 z^8 \\
& - 72 z^7 \alpha + 453 z^4 - 432 z^5 + 240 z^6 - 72 z^7 + 40 \alpha z^3 \beta_1 \\
& + 12 \alpha z^5 \beta_1 - 30 z^4 \alpha \beta_1 - 2 \alpha z^6 \beta_1 \left(\frac{\partial^2}{\partial z^2} f(z) \right) + (-15 \alpha z^2 \\
& - 4 z^2 + 49 z^3 \alpha + 2 z \alpha + 24 z^3 - 91 z^4 \alpha + 105 z^5 \alpha - 77 z^6 \alpha \\
& + z^9 - 9 z^8 \alpha - 9 z^8 + 35 z^7 \alpha - 61 z^4 + 85 z^5 - 70 z^6 + 34 z^7 \\
& + z^9 \alpha) \left(\frac{\partial^3}{\partial z^3} f(z) \right)
\end{aligned}$$

and the equation when the parameters are specialized:

```
> dneq:=gfun[algebraicsubs](bdeq,eq,f(z));
```

$$\begin{aligned}
dneq := & (9 z^4 - 36 z^3 + 61 z^2 - 50 z + 4) \left(\frac{\partial}{\partial z} f(z) \right) \\
& + (141 z^3 - 75 z^4 + 15 z^5 - 123 z^2 + 48 z - 6) \left(\frac{\partial^2}{\partial z^2} f(z) \right) \\
& + (42 z^4 - 48 z^3 - 18 z^5 + 3 z^6 + 27 z^2 - 6 z) \left(\frac{\partial^3}{\partial z^3} f(z) \right)
\end{aligned}$$

The linear recurrence satisfied by the d_n follows directly

```
> dnRec:=op(select(has,gfun[diffeqtoec](dneq,f(z),a(n)),n));
```

$$\begin{aligned}
dnRec := & (6 n^2 + 3 n^3) a(n) \\
& + (-93 n - 36 - 75 n^2 - 18 n^3) a(n+1)
\end{aligned}$$

$$\begin{aligned}
&+ (568 n + 404 + 267 n^2 + 42 n^3) a(n + 2) \\
&+ (-1193 n - 1176 - 411 n^2 - 48 n^3) a(n + 3) \\
&+ (1042 n + 1240 + 291 n^2 + 27 n^3) a(n + 4) \\
&+ (-78 n^2 - 336 n - 480 - 6 n^3) a(n + 5)
\end{aligned}$$

4. The integrals q_n satisfy a linear recurrence

The property above does not depend on the specific divergent series $\tilde{x}$ that one is resumming. This allows one to precompute efficiently the integrals $q_n(z)$ given the mapping function ϕ .

Indeed the general theory of holonomic function has recently led to symbolic summation and integration algorithms that turn out to apply to the integral representation of $q_n(z)$. The goal of these algorithms is to derive (systems of) linear functional equations, differential or difference, satisfied by a sum or an integral. We now proceed to use a prototypical implementation of them in the package [Mgfun](#) to obtain a recurrence on the $q_n(z)$. Then, we prove a theorem that by-passes the general theory of holonomic functions, and recover the same recurrence in a more direct way.

 (This section uses a version of [Mgfun](#) that is not distributed yet.)

```
[ > readlib (Mgfun) :
```

```

>  $\phi := \frac{1}{(1-t)^2} - 1$ 
>  $F := e^{\left(-\frac{\phi}{z}\right)} t^n \left( \frac{\partial}{\partial t} \phi \right)$ 
>  $ct := \text{Mgfun}_{\text{creative\_telescoping}}(F, n::\text{shift}, t::\text{diff})$ 
 $ct := [(z + n z) \_f(n, t) + (3 z + 3 n z) \_f(n + 2, t)$ 
 $+ (-n z - z) \_f(n + 3, t) + (-3 n z - 3 z - 2) \_f(n + 1, t),$ 
 $z t (-1 + 3 t - 3 t^2 + t^3) \_f(n, t)]$ 
>  $h[n](t) := \% [1];$ 
 $h_n(t) := (z + n z) \_f(n, t) + (3 z + 3 n z) \_f(n + 2, t)$ 
 $+ (-n z - z) \_f(n + 3, t) + (-3 n z - 3 z - 2) \_f(n + 1, t)$ 
>  $H[n](t) := \% \% [2];$ 
 $H_n(t) := z t (-1 + 3 t - 3 t^2 + t^3) \_f(n, t)$ 

```

The meaning of the previous computation is that the differential equation

$$h_n(t) + \left(\frac{\partial}{\partial t} H_n(t) \right) = 0$$

holds. This can be viewed as a differential-difference relation satisfied by the integrand $f_n(z, t)$. Now, integrating between -1 and 1 returns a non-homogeneous differential equation in the integral

$$q_n(z) = \int_{-1}^1 f_n(z, t) dt,$$

namely

$$\int_{-1}^1 h_n(t) dt + H_n(1) - H_n(-1) = 0,$$

where the integral rewrites in terms of $q_n(z)$ as

```
> Int(h[n](t), t=-1..1) = eval(subs(_f=unapply(q[n](z), n, t), ct[1]));
```

$$\int_{-1}^1 (z + n z) _f(n, t) + (3 z + 3 n z) _f(n + 2, t) + (-n z - z) _f(n + 3, t) + (-3 n z - 3 z - 2) _f(n + 1, t) dt$$

$$= (z + n z) q_n(z) + (3 z + 3 n z) q_{n+2}(z) + (-n z - z) q_{n+3}(z) + (-3 n z - 3 z - 2) q_{n+1}(z)$$

As to the non-homogeneous part $H_n(1) - H_n(-1)$, we readily evaluate it, verifying that it is 0 by chance.

```
> H[n](t) := factor(eval(subs(_f=unapply(F, n, t), H[n](t))));
```

$$H_n(t) := z t (-1 + t)^3 f$$

```
> assume(z>0); assume(n, integer);
factor(simplify(limit(op(2, %), t=1) - limit(op(2, %), t=-1))):
non_hom:=subs([z='z', n='n'], %); z:='z':
n:='n':
```

$$non_hom := 2$$

Consequently, we have obtained the following recurrence on the integrals $q_n(z)$.

```
> collect(eval(subs(_f=unapply(q[n](z), n,
t), ct[1])), q, factor)=0;
```

$$(-3nz - 3z - 2)q_{n+1}(z) - z(n+1)q_{n+3}(z) \\ + 3z(n+1)q_{n+2}(z) + z(n+1)q_n(z) = 0$$

```
> collect(subs(n=n-4,%), q, factor);
```

$$-z(n-3)q_{n-1}(z) + 3z(n-3)q_{n-2}(z) \\ + z(n-3)q_{n-4}(z) + (-3nz + 9z - 2)q_{n-3}(z) = 0$$

More generally, a differential equation with respect to z , or even a system of mixed differential-difference equations could be obtained by the same algorithms.

The following result had not been noticed by Lutz et al., but might prove useful in numerical computations.

Theorem. With the above notations, let

$$\sum_{k=0}^K p_k(n) a(n+k) = 0$$

be the linear recurrence satisfied by the Taylor coefficients at the origin of a power series solution of the first-order linear differential equation

$$\frac{\partial}{\partial t} G(t) = \left(\frac{\frac{\partial^2}{\partial t \partial t} \phi(t)}{\frac{\partial}{\partial t} \phi(t)} - \frac{\frac{\partial}{\partial t} \phi(t)}{z} \right) G(t).$$

Then the integrals q_n satisfy the recurrence

$$\sum_{k=0}^K p_k(-n) q_{n-k-1}(z) = 0.$$

Proof. The differential equation in the statement above is satisfied by the function

$$e^{\left(-\frac{\phi(u)}{z}\right)} \left(\frac{\partial}{\partial u} \phi(u) \right).$$

Since the integrals $q_n(z)$ rewrite as

$$\int_0^1 e^{\left(-\frac{\phi(u)}{z}\right)} u^n \left(\frac{\partial}{\partial u} \phi(u) \right) du,$$

by integration by parts and differentiation under the integral sign, they satisfy the announced linear recurrence.

The following one-line procedure computes a recurrence on the integral

$$q_n(z) = \int_0^\infty \mathbf{e}^{\left(-\frac{t}{z}\right)} \psi(t)^n dt$$

where ψ is the functional inverse of a rational function ϕ . It takes as input $\phi(t)$, t , g , n , z where all the arguments except the first one are symbols that appear in the output linear recurrence relating the $q_n(z)$.

```
> recqnofz := proc(
```

```

 $\phi::\text{ratpoly}, t::\text{name}, q::\text{name}, n::\text{name}, z::\text{name})$ 
local gf, a;
  op(select(has, eval(subs( $n = -n$ ,
     $a = \text{subs}(\_A = q, \text{proc}(x) \_A(-x - 1) \text{end}$ ),
    gfun[diffeqtorec](diff(gf(t), t)
      + (diff( $\phi, t$ ) /  $z - \text{diff}(\phi, t, t) / \text{diff}(\phi, t)) * \text{gf}(t), \text{gf}(t),$ 
      a(n))))), n))
end

```

Example: $\phi = \frac{1}{(1-t)^2} - 1$

```
> qnRec:=recqnofz(1/(1-t)^2-1,t,q,n,z);
```

```
qnRec := (-n z + 3 z) q(n-1) + (-9 z + 3 n z) q(n-2)
        + (-3 n z + 9 z - 2) q(n-3) + (n z - 3 z) q(n-4)
```

We have obtained the same recurrence as when using [Mgfun](#).

```
> qnRec:=applyop(factor,[2,2],applyop(collec
t,[2,1],readlib(isolate)(subs(n=n+1,qnRec)
,q(n)),q,normal));
```

```
qnRec := q(n) = -((6 z - 3 n z) q(n-1)
        + (-6 z + 3 n z + 2) q(n-2) + (2 z - n z) q(n-3)) / (
        z(n-2))
```

5. Numerical experiments

We know the recurrence followed by the q_n , and we now have to compute initial conditions. By applying the change of variable $t = \phi(u)$, we obtain

$$q_n = \int_0^1 e^{\left(-\frac{\phi(u)}{z}\right)} u^n \left(\frac{\partial}{\partial u} \phi(u) \right) du$$

which gives the following initial condition

```
> phi:=subs(z=u,solve(eq,f));
```

$$\phi := -\frac{u(-2+u)}{(u-1)^2}$$

```
> assume(z>0):
```

```
> q0:=int(exp(-phi/z)*diff(phi,u),u =0..1);
```

$$q0 := z \sim$$

```
> q0:=subs(z='z',q0); z:='z':
```

$$q0 := z$$

Note that this initial value is a posteriori obvious.

Thus we now have both recurrence and initial condition. The solution q_n to the recurrence equation `qnRec` is a dominated solution, which means that any numerical error grows exponentially. To avoid this, we run the recurrence backwards from any non trivial initial conditions. The dominating solution disappears quickly, and we obtain the solution q_n because when the recurrence is run backwards it becomes a dominating solution. We therefore add a parameter *NN* indicating from where we start running backwards.

```
> eval(collect(op(2,isolate(subs(n=n+3,qnRec),q(n))),q,normal),q=proc(n) q(n,z,NN)
end);
```

$$q(n+3, z, NN) - 3 q(n+2, z, NN) \\ + \frac{(3z + 3nz + 2) q(n+1, z, NN)}{z(n+1)}$$

```
> qnprocrev := subs(_REC = %, q = qnprocrev, proc(n, z, NN)
option remember;
    if n = NN then 0
    elif n = NN - 1 then 0
    elif n = NN - 2 then 1
    else _REC
    fi
end)
```

```
qnprocrev := proc(n, z, NN)
```

```
option remember;
    if n = NN then 0
    elif n = NN - 1 then 0
    elif n = NN - 2 then 1
    else qnprocrev(n+3, z, NN) - 3*qnprocrev(n+2, z, NN)
        + ((3*z + 3*n*z + 2)*qnprocrev(n+1, z, NN)) / (
            z*(n+1))
    fi
end
```

Here is a procedure to compute q_n numerically.

```
> `evalf/q/digits`:=0:
```

```

> evalf/q := proc(n, z, NN)
  global evalf/q, 'evalf/q/digits';
  option remember;
  if 'evalf/q/digits' < Digits then
    'evalf/q/digits' := Digits;
    evalf/q := subsop(4 = NULL, op(evalf/q))
  fi;
  if n = NN then 0
  elif n = NN - 1 then 0
  elif n = NN - 2 then 1.0
  else procname(n + 3, z, NN) + (2 + 3*z + 3*z*n)*
    procname(n + 1, z, NN) / (z*(n + 1))
    - 3*procname(n + 2, z, NN)
  fi
end
> dnRec :=
  gfundiffeqtorec( { dneq, D(f)(0) = 1, f(0) = 0 }, f(z), a(n))
> dnproc := gfun[rectoproc](dnRec, a(n), remember) :

```

Here is a procedure to compute d_n numerically.

```

> `evalf/d/digits`:=0:
> evalf/d := proc(n)
  global evalf/d, 'evalf/d/digits';
  option remember;
  if 5 < n then
    if 'evalf/d/digits' < Digits then

```

```

        'evalf/d/digits' := Digits;
        evalf/d := subsop(4 = NULL, op(evalf/d))
    fi;
    evalf(− (− 225*procname(n − 5)
        − 70*procname(n − 1)
        + 804*procname(n − 4)
        − 1011*procname(n − 3)
        + 514*procname(n − 2) + (
        165*procname(n − 5) − 693*procname(n − 4)
        + 1048*procname(n − 3)
        − 683*procname(n − 2)
        + 157*procname(n − 1) + (
        − 39*procname(n − 5) + 195*procname(n − 4)
        − 363*procname(n − 3)
        + 309*procname(n − 2)
        − 114*procname(n − 1) + (
        − 18*procname(n − 4) + 42*procname(n − 3)
        − 48*procname(n − 2) + 3*procname(n − 5)
        + 27*procname(n − 1))*n)*n)*n) / (
        (−6 + (12 − 6*n)*n)*n))
    elif n = 0 then 0
    elif n = 1 then 1
    elif n = 2 then evalf(1 / 3)
    elif n = 3 then evalf(− 23 / 108)
    else evalf(− 2749 / 3888)
    fi

```

end

Finally, the following procedure computes values of the double confluent Heun function as follows. First, an upper value of NN is selected and $2\,NN$ is used in the backward recurrence to compute the scaling to use for $q_n(z)$ in view of the actual initial condition. Then the summation is performed up to the NN th term. If the relative error of the last term is larger than $10^{(-Digits)}$, then NN is doubled and the computation starts again. Note that option `remember` has been used in *qnprocrev* so as to avoid duplicating some of the work.

```
> time(evalf(q(0,10.2,3000),21));  
                                     .399  
> valheun:=subs(_q0=q0,proc(z)  
    local tot,i,N,NN,lambda,st,D;  
    N:=10;  
    st:=time();  
    do  
        N:=floor(2*N);  
        NN:=N+floor(sqrt(N))+10;  
        D:=Digits+3*ilog10(N) +floor(log(N));  
        lambda:=_q0/evalf(q(0,z,NN),D);  
  
        tot:=add(evalf(d(i),D)*evalf(q(i,z,NN),D),  
            i=1..N)*lambda;  
        if  
            abs(evalf(d(N),D)*evalf(q(N,z,NN),D)*lambda)  
            <abs(tot)*10^(-Digits) then break fi;
```

```

od;
userinfo(1, valheun, "N=", N, "z=", 1.0*z, "time
:", time()-st, "digits:", D);
tot;
end):

```

We print information on the number of terms and time used for each point in a plot of the solution from 0 to 15:

```


> infolevelvalheun := 1
> plot(valheun, 0..50);
valheun:  "N="    80    "z="    1.089857709    "time:"
.141    "digits:"    17
valheun:  "N="   160    "z="    2.038137074    "time:"
.270    "digits:"    21
valheun:  "N="   320    "z="    3.104576397    "time:"
.579    "digits:"    21
valheun:  "N="   320    "z="    4.178084772    "time:"
.571    "digits:"    21
valheun:  "N="   320    "z="    5.246490950    "time:"
.599    "digits:"    21
valheun:  "N="   640    "z="    6.237040242    "time:"
1.221    "digits:"    22
valheun:  "N="   640    "z="    7.262696659    "time:"
1.080    "digits:"    22
valheun:  "N="   640    "z="    8.323431992    "time:"
1.129    "digits:"    22
valheun:  "N="   640    "z="    9.380765534    "time:"
1.240    "digits:"    22
valheun:  "N="   640    "z="   10.46836305    "time:"
1.201    "digits:"    22
valheun:  "N="  1280    "z="   11.42631952    "time:"
2.520    "digits:"    26
valheun:  "N="  1280    "z="   12.50475163    "time:"
2.369    "digits:"    26

```

valheun:	"N="	1280	"z="	13.58761188	"time:"
2.601	"digits:"	26			
valheun:	"N="	1280	"z="	14.63114838	"time:"
2.750	"digits:"	26			
valheun:	"N="	1280	"z="	15.57877949	"time:"
2.800	"digits:"	26			
valheun:	"N="	1280	"z="	16.70560455	"time:"
3.110	"digits:"	26			
valheun:	"N="	1280	"z="	17.66017317	"time:"
3.219	"digits:"	26			
valheun:	"N="	1280	"z="	18.77056341	"time:"
3.340	"digits:"	26			
valheun:	"N="	2560	"z="	19.75344692	"time:"
7.290	"digits:"	26			
valheun:	"N="	2560	"z="	20.83182591	"time:"
8.091	"digits:"	26			
valheun:	"N="	2560	"z="	21.85869773	"time:"
8.440	"digits:"	26			
valheun:	"N="	2560	"z="	22.93013074	"time:"
9.180	"digits:"	26			
valheun:	"N="	2560	"z="	23.91404071	"time:"
9.710	"digits:"	26			
valheun:	"N="	2560	"z="	24.97532053	"time:"
10.209	"digits:"	26			
valheun:	"N="	2560	"z="	26.07769200	"time:"
10.941	"digits:"	26			
valheun:	"N="	2560	"z="	27.03730960	"time:"
11.229	"digits:"	26			
valheun:	"N="	2560	"z="	28.07372290	"time:"
11.961	"digits:"	26			
valheun:	"N="	2560	"z="	29.14443926	"time:"
12.589	"digits:"	26			
valheun:	"N="	2560	"z="	30.19192632	"time:"
12.810	"digits:"	26			
valheun:	"N="	2560	"z="	31.20542392	"time:"

13.741	"digits:"	26			
valheun:	"N="	5120	"z="	32.33074020	"time:"
20.060	"digits:"	27			
valheun:	"N="	5120	"z="	33.34188693	"time:"
12.880	"digits:"	27			
valheun:	"N="	5120	"z="	34.42150189	"time:"
15.039	"digits:"	27			
valheun:	"N="	5120	"z="	35.39979350	"time:"
17.241	"digits:"	27			
valheun:	"N="	5120	"z="	36.46932464	"time:"
19.520	"digits:"	27			
valheun:	"N="	5120	"z="	37.47567008	"time:"
21.760	"digits:"	27			
valheun:	"N="	5120	"z="	38.52759104	"time:"
24.110	"digits:"	27			
valheun:	"N="	5120	"z="	39.55603605	"time:"
26.450	"digits:"	27			
valheun:	"N="	5120	"z="	40.63272267	"time:"
28.740	"digits:"	27			
valheun:	"N="	5120	"z="	41.66969985	"time:"
31.479	"digits:"	27			
valheun:	"N="	5120	"z="	42.73015878	"time:"
33.841	"digits:"	27			
valheun:	"N="	5120	"z="	43.78183659	"time:"
36.099	"digits:"	27			
valheun:	"N="	5120	"z="	44.74821926	"time:"
39.000	"digits:"	27			
valheun:	"N="	5120	"z="	45.85580287	"time:"
40.981	"digits:"	27			
valheun:	"N="	5120	"z="	46.84643885	"time:"
43.800	"digits:"	27			
valheun:	"N="	5120	"z="	47.90266342	"time:"
45.850	"digits:"	27			
valheun:	"N="	5120	"z="	48.91360511	"time:"
48.260	"digits:"	27			

```
valheun:  "N="  5120  "z="  50.0  "time:"  50.6  
99  "digits:"  27  
valheun:  "N="  160  "z="  1.563997392  "time:"  
1.450  "digits:"  21
```



This curve is to be contrasted with the irregular plot we got from the same series using summation to the least term.

Sections 1 and 2 of this worksheet only depend on the Heun

differential equation, and can easily be adapted to any linear differential equation. Sections 3 and 4 compute the recurrences satisfied by the coefficients q_n and d_n , which depend on the choice of the mapping function ϕ only for q_n , and on the differential equation as well for d_n . Section 5 details the numerical computations. It depends on ϕ and on the recurrences found in the previous sections. This worksheet can be adapted to another mapping function and to another linear differential equation.

References

- [Lutz et al.] *On the convergence of Borel approximants*, by W. Balser, D. A. Lutz and R. Schäfke, (2000). Preprint.
- [DuLoRi92] *Kovacic's Algorithm and Its Application to Some Families of Special Functions*, by Anne Duval and Michèle Loday-Richaud, *Applicable Algebra in Engineering, Communication and Computing*, (1992), vol. 3, p. 211-246.
- [Durand] *On the convergence of Borel approximants [summary of a talk by Donald Lutz]*, by Marianne Durand, (2001). To appear in: *Algorithms Seminar, 2000-2001*, INRIA Research Report.